



ODS: Output Delivery System Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

ODS allows you to format your reports in various formats such as HTML, PDF, RTF, Microsoft Excel, and many others. It enables you to customize your reports by selecting only the results you want to see. It even lets you apply styles to your reports including many supplied by SAS[®] or those that you create. This tip sheet presents the most common statements and options in ODS.

Output Destination

Basic Usage

```
ods output output-object-name=data-set-name;
... procedure code ...
ods output close;
```

Use **ods trace on;** to determine the output object names.

Customizing Reports

ods proclabel "text";
Specifies text to use in place of the procedure label in the table of contents.

ods proctitle; or **ods noproctitle;**
Specifies whether or not the procedure title should be printed.

ods select selection(s);
ods exclude selections(s);
Specifies the output objects to show or exclude from the report, respectively.

ods show;
Displays current select and exclude lists.

ods trace on;
Displays information about output objects as they are created. This information includes the name of the object, the name of the template, labels, paths, and label paths.

Locating Style Definitions

odstemplates
Entering **odstemplates** in the command bar opens the Template Browser. The styles supplied by SAS are in the SASHELP.TMPLMST/Styles folder.

proc template; list styles; run;
Lists the SAS-supplied styles to the log.

Example

```
* Open several ODS destinations;
ods html frame="report.html"
      file="reportb.html"
      contents="reportc.html"
      style=styles.analysis;
ods pdf file="report.pdf" notoc;
ods rtf file="report.rtf" bodytitle;
ods tagsets.excelxp file="report.xls";
ods output variables=work.vardata;

* Select only the "variables" table;
ods select variables;

* Turn off proc titles;
ods noproctitle;

* Run the Contents Procedure;
proc contents data=sashelp.class; run;

* Close all ODS destinations;
ods _all_ close;

* Re-open the listing destination;
ods listing;
```

The SAS System

Alphabetic List of Variables and Attributes			
#	Variable	Type	Len
3	Age	Num	8
4	Height	Num	8
1	Name	Char	8
2	Sex	Char	1
5	Weight	Num	8

Advanced Techniques

For more information on advanced techniques such as the **DOCUMENT** destination which allows extensive manipulation of reports, or **PROC TEMPLATE** which allows you to modify styles as well as create table templates or modify table templates used by SAS procedures, see <http://support.sas.com/v9doc>.



SAS Institute Inc. World Headquarters
+1 919 677 8000 www.sas.com/offices

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2008, SAS Institute Inc. All rights reserved. 479398_01.08

Common Destinations

“Destination” is a term for the ODS driver that generates a specific output format.

Destination Name	Description
Listing	Plain text
HTML	Primary format used on the World Wide Web (WWW)
XML	Plain text data interchange format
PDF	Works well for printing as well as on-screen viewing
Postscript	Works well for printing
RTF	Used by most word processors
Tagsets.ExcelXP	Share SAS data with Microsoft Excel
Output	Creates SAS data sets from ODS tables

Opening and Closing Destinations

ods destination-name <option(s)>;
Opens the ODS destination, *destination-name*, with options, *option(s)*.

ods destination-name close;
Closes the ODS destination, *destination-name*.

ods _all_ close;
Closes all ODS destinations (including Listing).

Basic ODS Usage

```
ods destination-name;
... procedure code ...
ods destination-name close;
```

HTML Destination

Basic Usage

```
ods html <option(s)>;
... procedure code ...
ods html close;
```

File Options

file=“filename” or **body=“filename”**
Specifies the name of the file that will contain the output tables.

contents=“filename”
Specifies the name of the file that will contain a table of contents for the output.

frame=“filename”
Specifies the name of the file that includes the body file and table of contents into a set of scrollable frames.

stylesheet=“filename”
Specifies an external filename for styles.

The following file sub-options can be specified in parentheses following each of the above options.

no_top_matter
no_bottom_matter
Specifies that no beginning or ending markup, respectively, should be inserted into the output file.

url=“URL”
Specifies a URL to be used in place of the filename in links to the file.

Other Commonly Used Option

style=style-definition
Specifies the style definition to use.

PDF Destination

Basic Usage

```
ods pdf <option(s)>;
... procedure code ...
ods pdf close;
```

File Options

file=“filename”
Specifies the name of the file that will contain the output tables.

Document Metadata Options

author=“text”
keywords=“text”
subject=“text”
title=“text”
Specifies the author, keywords, subject, and title in the metadata of the PDF document.

Other Commonly Used Options

columns=n
Specifies the number of columns on each page.

compress=n
Specifies the level of compression.

notoc
Specifies that the table of contents bookmarks should not be generated.

startpage=yes | no | now
Controls page breaks.

style=style-definition
Specifies the style definition to use.

uniform
Specifies that multiple page tables should retain their width across pages.

RTF Destination

Basic Usage

```
ods rtf <option(s)>;
... procedure code ...
ods rtf close;
```

File Options

file=“filename”
Specifies the name of the file that will contain the output tables.

Document Metadata Options

author=“text”
operator=“text”
title=“text”
Specifies the author, operator, and title in the metadata of the RTF document.

Other Commonly Used Options

bodytitle
Specifies that the titles and footnotes should appear in the body of the report, not as RTF instructions.

columns=n
Specifies the number of columns on each page.

sasdate
Specifies that the time and date that the SAS program was submitted should be written to the RTF, instead of the time that the file was opened.

startpage=yes | no | now
Controls page breaks.

style=style-definition
Specifies the style definition to use.